

Contents

Program ZVMXPDFARCHIVE	2
Program Layout.....	2
Program Logic	5
Selection Screen Validation	5
Select Data	5
ArchiveLink Configuration Check	7
Process data	7
Run Archive out Routine	9
Show Results in ALV	10
User Command	11

Program ZVMXPDFARCHIVE

Transaction: ZMXARCH

Program Layout

- Initial Screen: These are the screen fields parameters:

The screenshot shows a SAP selection screen titled "Mexico E-carm Intervention - OM Documents". It contains a "Selection Parameters" section with the following fields:

Field	Value	Field	Value
Company Code			
Output Type	<input checked="" type="checkbox"/>		
Billing Document		to	
Billing Type		to	
Billing Date		to	














Below the fields is a checkbox labeled "Rewrite Existing Archived Docs" which is currently unchecked. To the right of the date range fields, there are three yellow folder icons stacked vertically.

- Execution Mode: We have two modes.
 - Archive Documents: Archiving only documents that are not archived yet;
 - It is also mandatory fill at least Billing document or Billing Date field for performance reasons.
 - Rewrite Existing Archived Docs: Delete old archive PDF versions and create a new one.
 - It is also mandatory fill at Least Billing document for Performance reasons.
 - When the user press F8 or click on run will appear a message informing the user that the program is running in Rewrite mode.



- Refresh Button on ALV screen: For performance reasons the Archiving is madden in background. So, The final report cannot reflect the real archiving result at the first time. For this reason I created a new button I create a new functionality called "Refresh". Once the user execute the

Mexico E-arm Intervention - OM Documents									
Processing log Refresh Display Archiving List									
Billing Doc.	Co...	Sal...	Ms...	Bill...	Delreqs.	Doc. type	Document ID
970000933	376	MX00	01	01	ZHXF	F2	1		
									Description
									Archiving Process Started. Please wait a few minutes and click on Refresh button

Mexico E-arm Intervention - OM Documents											
											
Processing log Refresh Display Archiving List											
Billing Doc.	Co...	Sal...	Ms...	Bill...	Delreqs.	Doc. type	...	Document ID	Description
 970000933	376	MX00	01	01	ZMXF	F2	1	ZSD8IL376P	MD	6C3BE5A99C721EE489F02800D533E615	Document Archived

-

SAP E-carm Intervention - OM Documents

Billing Doc.	Co...	Sal...	Ms...	Bill...	Del.reqs.	Doc. type	...	Document ID
970000933	376	MX00	01	01	ZMXF	F2	0	ZSDBIL376P	MD	6C3BE5A99C721EE489F02800D533E615

Output Processing analysis

Type	Message text
<input checked="" type="checkbox"/>	Object 0970000933
<input checked="" type="checkbox"/>	Output type: MX DomCustInvoiceCOP
<input checked="" type="checkbox"/>	Processing log for program ZVXX_OE_INV routine ENTRY_SMARTFORM
<input checked="" type="checkbox"/>	Message output with spool request number 155100

- Display Archiving List: This option will show all files archived via ECC. It will show the same files of OAAD.

Mexico E-carm Intervention - OM Documents

Processing log Refresh Display Archiving List

Billing Doc.	Co...	Sal...	...	Ms...	Bill....	Del.reqs.	Doc. type	...	Document ID	Description
970000933	376	MX00	01 01	ZMXF	F2	0	ZSDBIL376P	MD	6C3BE5A99C721EE489F02B00D533E615	Document Already Archived!

Attachment List

AttachmentFor0970000933

Icon	Title	Creator Name	Created On
	Mexico pdf archiving OM document t...	MILLER NUNES	20.04.2015

Program Logic

Selection Screen Validation

Form F_SCREEN_PARAMETERS_VALIDATION

Before select the documents, the program validates the screen parameters.

1. If the parameter Rewrite Existing Archived Docs(P_DELETE) is not initial and no Billing Document was entered (S_VBELN initial) show the message below and stop program execution:
"For Rewrite option Document Number is Mandatory!"
2. If P_DELETE is not initial and any Billing Document was entered, call the FM POPUP_TO_CONFIRM in order to show the message: **"Rewrite option was flagged. Do you really want to delete the existing archiving and create a new one?"**. If the user click in YES continue the program, otherwise stop program execution.
3. If Billing Document and Billing date was in blank show the message below and stop the program execution:
"Please fill at least Billing Document or Billing Date!"

Select Data

Form F_SELECT_DATA

1. We have to select Archiving document type associated with the output that we want to archive in order to validate if the documents was already archived or not. The Archiving document type is stored on table T685B field ARCOBJART. See the logic below:

```
SELECT SINGLE
      KAPPL
      KSCHL
      ARCOBJART
FROM T685B
INTO E_T685B
WHERE KAPPL EQ C_V3
      AND KSCHL EQ P_KSCHL.

*If no output was found show message error:
*" Output & is not defined
IF SY-SUBRC IS NOT INITIAL.
  MESSAGE I024(VN) WITH P_KSCHL.
  STOP.
ENDIF.

IF E_T685B-ARCOBJART IS INITIAL.
* If no Archive Document type was configured show message error:
* "Maintain archiving parameter(s) for output type &1 (appl. &2)
  MESSAGE I096(VN) WITH P_KSCHL C_V3.
  STOP.
ENDIF.
```

2. Then, we should validate if the Archive Document Type exists by selecting it on ArchiveLynk Table TOADV.

```
SELECT SINGLE
      AR_OBJECT
      DOC_TYPE
FROM   TOADV
INTO   E_TOADV
WHERE  AR_OBJECT EQ E_T685B-ARCOBJART.
```

3. Select Billing Document Header (VBRK Table) according to screen parameters

```
SELECT VBELN
      FKART
      VKORG
      VTWEG
      FKDAT
      BUKRS
      KUNRG
      SPART
FROM   VBRK
INTO   TABLE TS_VBRK
WHERE  VBELN IN S_VBELN
      AND FKART IN S_FKART
      AND FKDAT IN S_FKDAT
      AND BUKRS EQ P_BUKRS.
```

**If no billing document was found show message error:*

```
IF SY-SUBRC IS NOT INITIAL.
```

** No billing documents selected*

```
MESSAGE I805(VF).
```

```
STOP.
```

```
ENDIF.
```

4. Get Bill-To party from Billing document (VBPA table)

** Get Bill-to party*

```
SELECT VBELN
      KUNNR
FROM   VBPA
INTO   TABLE TS_VBPA
      FOR ALL ENTRIES IN TS_VBRK
WHERE  VBELN EQ TS_VBRK-VBELN
      AND POSNR EQ C_000000
      AND PARVW EQ C_RE.

IF SY-SUBRC IS INITIAL.
  SORT TS_VBPA BY VBELN ASCENDING.
ENDIF.
```

ArchiveLink Configuration Check

Form F_VALIDATE_ARCHIVELINK_CONF

Call FM ARCHIV_APPLICATIONINFO_GET to validate the configuration of Archive system:

```
* Get/Check Archiving information
CALL FUNCTION 'ARCHIV_APPLICATIONINFO_GET'
  EXPORTING
    AR_OBJECT      = E_T685B-ARCOBJART
    SAP_OBJECT     = C_VBRK
  IMPORTING
    ARCHIV_ID      = W_ARCHIV_ID
  EXCEPTIONS
    ERROR_COMMUNICATIONTABLE = 1
    ERROR_CONNECTIONTABLE   = 2
    OTHERS                 = 3.

* If no ArchiveLink configuration was found show message error
IF SY-SUBRC NE 0.
* "SAP ArchiveLink: Obj. type not assigned to
* storage syst.(Customizing) & & "
  MESSAGE ID SY-MSGID
           TYPE C_S
           NUMBER SY-MSGNO
           WITH SY-MSGV1
                SY-MSGV2
                SY-MSGV3
                SY-MSGV4.

  STOP.

* Implement suitable error handling here
ENDIF.
```

Process data

Form F_PROCESS_DATA

For each document selected:

1. Get Archive information: We should call the FM ARCHIV_CONNECTINFO_GET_META (that are inside of FORM F_GET_ARCHIVE_DATA) to retrieve Archiving information passing OBJECT_ID = VBRK-VBELN and SAP_OBJECT = 'VBRK'. Then delete Archive files that does not have the same Archive Document Type. Each Archive Document Type means a different type of archive file such as Invoice PDF, Fiscal XML and so on.

```
CALL FUNCTION 'ARCHIV_CONNECTINFO_GET_META'
  EXPORTING
    OBJECT_ID      = US_OBJECT_ID
    SAP_OBJECT     = US_SAP_OBJECT
  IMPORTING
    COUNT          = US_COUNT
  TABLES
    CONNECT_INFO   = TA_CONNECTIONS
  EXCEPTIONS
    ERROR_CONNECTIONTABLE = 1
    OTHERS           = 2.
```

```

IF NOT SY-SUBRC EQ 0.
    FREE TA_CONNECTIONS.
ELSE.
    * Check only documents that have the Same Archive Document type of Out
    put
    DELETE TA_CONNECTIONS WHERE AR_OBJECT NE US_ARCOBJART.
ENDIF.

```

2. If at least one Archive File was found for the same Archive Document Type (TS_CONNECTION is not initial)
 - a. If P_DELETE is not initial: run the archive deletion logic (form F_DELETE_ARCHIVE) .
The program uses Method CL_ALINK_CONNECTION=>DELETE_BY_KEY and FM ARCHIVOBJECT_DELETE to delete the archive files.

```

* Delete all PDF documents related to the output
LOOP AT TA_CONNECTIONS
    ASSIGNING <FS_CONNECTIONS>.

    CALL METHOD CL_ALINK_CONNECTION=>DELETE_BY_KEY
    EXPORTING
        MANDT      = <FS_CONNECTIONS>-MANDT
        SAP_OBJECT = <FS_CONNECTIONS>-SAP_OBJECT
        OBJECT_ID  = <FS_CONNECTIONS>-OBJECT_ID
        ARCHIV_ID  = <FS_CONNECTIONS>-ARCHIV_ID
        ARC_DOC_ID = <FS_CONNECTIONS>-ARC_DOC_ID
    EXCEPTIONS
        NOT_FOUND = 1
        OTHERS    = 2.

    IF SY-SUBRC NE 0.
        * Implement suitable error handling here
        CLEAR SY-SUBRC.
    ENDIF.

    CALL FUNCTION 'ARCHIVOBJECT_DELETE'
    EXPORTING
        DOC_TYPE           = E_TOADV-DOC_TYPE
        ARCHIV_DOC_ID      = <FS_CONNECTIONS>-ARC_DOC_ID
        ARCHIV_ID          = <FS_CONNECTIONS>-ARCHIV_ID
        SAP_OBJECT         = <FS_CONNECTIONS>-SAP_OBJECT
        AR_OBJECT          = <FS_CONNECTIONS>-AR_OBJECT
        OBJECT_ID          = <FS_CONNECTIONS>-OBJECT_ID
    EXCEPTIONS
        ERROR_ARCHIV      = 1
        ERROR_COMMUNICATIONTABLE = 2
        ERROR_KERNEL      = 3
        OTHERS            = 4.

    IF SY-SUBRC NE 0.
        * Implement suitable error handling here
        CLEAR <FS_CONNECTIONS>-ARC_DOC_ID.
    ENDIF.
ENDLOOP.

```


- b. If P_DELETE is initial: Select the last message output related with the current Billing document on NAST and insert it on IS_NAST internal table (form F_SELECT_NAST_MESSAGE) and insert the document on output table (TS_OUTPUT_RW) with the message 'Document Already Archived!' and Red Light sign (form F_FILL_OUTPUT_ARCHIVED_ERROR).

3. Run Archive logic.

Run Archive out Routine

Form F_RUN_ARCHIVE_OUTPUT

Once the output has the Archive configuration done and well formatted, the best choice to archive the PDF is create a new output only with archiving option. Therefore, the standard will do the job by itself.

```
** Fill NAST
MOVE: SY-MANDT      TO NAST-MANDT,
      C_V3          TO NAST-KAPPL,
      US_VBRK-VBELN TO NAST-OBJKY,
      SY-DATUM      TO NAST-ERDAT,
      SY-UNAME      TO NAST-USNAM,
      SY-UZEIT      TO NAST-ERUHR,
      US_VBRK-KUNRG TO NAST-PARNR,
      C_BP          TO NAST-PARVW,
      P_KSCHL       TO NAST-KSCHL,
      C_EN          TO NAST-SPRAS,
      1             TO NAST-NACHA,
      1             TO NAST-VSZTP,
      C_VBRK        TO NAST-OBJTYPE,
*      w_printer     TO nast-ldest,
      C_LOCL        TO NAST-LDEST,
**      nast-dimme = 'X' => print immediately
*      c_x           TO nast-dimme,
      TEXT-T02      TO NAST-TDCOVTITLE,
      2             TO NAST-TDARMOD.

READ TABLE TS_VBPA
ASSIGNING <FS_VBPA>
  WITH KEY VBELN = US_VBRK-VBELN
  BINARY SEARCH.

IF <FS_VBPA> IS ASSIGNED
AND SY-SUBRC IS INITIAL.

  MOVE <FS_VBPA>-KUNNR TO NAST-PARNR.

  UNASSIGN <FS_VBPA>.

ENDIF.

APPEND NAST TO IS_NAST_TMP.

* Clear NAST Memory
CALL FUNCTION 'RV_MESSAGES_REFRESH'
```

```

EXPORTING
  MSG_KAPPL = C_V3.

* Insert NAST entry on buffer tables
CALL FUNCTION 'RV_MESSAGES_INSERT'
  TABLES
    TAB_XNAST = IS_NAST_TMP.

* insert NAST on database
CALL FUNCTION 'RV_MESSAGES_UPDATE'
  EXPORTING
    MSG_KAPPL           = C_V3
    MSG_NO_UPDATE_TASK = C_X
    MSG_OBJKY           = NAST-OBJKY.

COMMIT WORK.

*** Call Print Program
PERFORM EINZELNACHRICHT_DIALOG IN PROGRAM RSNAST00 USING W_SUBRC.
**
COMMIT WORK.
* add nast to internal table
APPEND NAST TO IS_NAST.

```

If print process was executed properly get archive data and show message of success. To do so, we need to check if the document was archived after output had being finished by call the Form F_GET_ARCHIVE_DATA again.

For performance issues the SAP can take a long finished the archive process for each document. So:

- 1) If IS_CONNECTIONS_NEW is not initial: insert the document on output table (TS_OUTPUT_RW) with the message 'Document Already Archived!' and Green Light sign.
- 2) Otherwise: insert the document on output table (TS_OUTPUT_RW) with the message 'Archive not completed yet. Review Output Log and/or Archive List before run the transaction again!' and Yellow Light sign.
- 3) If some error occurs during Output execution: insert the document on output table (TS_OUTPUT_RW) with the message 'An Error occurred during Output execution. Please Review Log!' and Red Light sign.

Show Results in ALV

Form F_SHOW_OUTPUT_RW

Build the ALV using the class CL_SALV_TABLE. Register Event ON_USER_COMMAND (Form F_HANDLE_USER_COMMAND) and show ALV

User Command

Form F_HANDLE_USER_COMMAND

Gui Status: SALV_TABLE_STANDARD

For this program, we have three custom buttons. Those buttons are included on Gui Status SALV_TABLE_STANDARD and should be handle by FORM F_HANDLE_USER_COMMAND.

Items 1 - 7							
Items 8 - 14							
Items 15 - 21					LOG Proces	REFRESH Refres	VIEW &y' Displa
Items 22 - 28							
Items 29 - 35							

- Processing Log: this option has the same logic of the button Processing Log out output Visualization in VF03->Go To->Header->Output.
To show the Popup with Log Messages we should call the FMs WFCM_PROTOCOL_REFRESH, WFCM_PROTOCOL_LOAD and WFCM_PROTOCOL_LOAD. Please see the Form F_SHOW_OUTPUT_PROTOCOL for more details.
- Display Archiving List: This option has the same functionality of OAAD transaction. It will show the Archived Document Files. Please see the Form F_VIEW_ARCHIVING for more details.
- Refresh: as I told before, for performance issues, the Archiving process is triggered in background, so at the moment that the ALV is showed some documents are not completely archived. The refresh button will check again if the archiving process was finished and will retrieve the ALV information. Please see the Form F_REFRESH for more details.